Task

Create a function which takes in a variable number of arguments and prints each one on its own line.

Also show, if possible in your language, how to call the function on a list of arguments constructed at runtime.



Functions of this type are also known as Variadic Functions.



Related task

   Call a function


import std.stdio, std.algorithm;
 
void printAll(TyArgs...)(TyArgs args) {
    foreach (el; args)
        el.writeln;
}
 
// Typesafe variadic function for dynamic array
void showSum1(int[] items...) {
    items.sum.writeln;
}
 
// Typesafe variadic function for fixed size array
void showSum2(int[4] items...) {
    items[].sum.writeln;
}
 
void main() {
    printAll(4, 5.6, "Rosetta", "Code", "is", "awesome");
    writeln;
    showSum1(1, 3, 50);
    showSum2(1, 3, 50, 10);
}